# Perl によるプログラミング入門 in 東海

## 第一回

## 2016/02/20 第一版

## 目 次

1	はじめまして	3
	1.1 はじめに	3
2	Perl について	4
	2.1 Perl 言語	4
3	最初のプログラム	4
	3.1 プログラムを書いて,実行してみる	4
	3.1.1 実行例	4
	3.1.2 01_hello.pl	5
	3.2 解説 - 周辺	5
	3.3 解説 - プログラムファイル	5
	3.4 練習問題	6
4	まずはプログラミングに慣れる	7
	4.1 簡単な計算	7
	4.1.1 02 calc.pl	7
	- 4.2 解説 - 値 と変数	7
	4.3 単語 - 値,変数	8
		8
		8
		9
		9
		0
5	今回目指すもの 1	0
		_
	5.1 今回目指すもの	U

6	TOI	OO リストを保存する	10
	6.1	$03\_init\_todo.pl\ldots\ldots\ldots\ldots\ldots$	10
	6.2	open() / close() 関数	11
	6.3	print() 関数	12
	6.4	練習問題	12
7	TOI	OO を追加する	12
	7.1	$04\_add\_todo.pl  .  .  .  .  .  .  .  .  .  $	12
	7.2	行入力演算子	13
	7.3	スカラの自動変換	13
	7.4	chomp() 関数	14
	7.5	練習問題	14
8	現在	の TODO を表示する	14
	8.1	$05\_list\_todo.pl \ . \ . \ . \ . \ . \ . \ . \ . \ . \ $	14
	8.2	defind() とファイルの終端	15
	8.3	while $\dot{\Sigma}$	15
	8.4	ブーリアン演算子	16
	8.5	練習問題	16
9	TOI	<b>OO</b> の内容のうち未完了のものだけ表示する	17
	9.1	$06\_list\_notyet\_todo.pl \ \dots \dots \dots \dots \dots \dots \dots$	17
	9.2	リスト / リスト変数	17
	9.3	$\operatorname{split}()$ 関数	18
	9.4	スカラコンテキスト / リストコンテキスト	18
	9.5	練習問題	19
10	тоі	OO を完了済みにする	19
	10.1	$07\_done\_todo.pl\ .\ .\ .\ .\ .\ .$	20
	10.2	if $\dot{\chi}$	20
	10.3	練習問題	21
11	まと	め	21
	11.1	準備	21
	11.2	基礎文法	21
	11.3	スカラ値	22
	11.4	リスト値	23
	11.5	変数	23
		11.5.1 スカラ変数	23
		11.5.2 リスト変数	23
	11 6	<b>油質</b> 了.	24

	11.7 関数	24
	11.8 while 文	25
	11.9 if 文	25
	11.10コンテキスト	25
<b>12</b>	練習問題の答え	26
	12.1 3. 練習問題	26
	12.2 4. 練習問題	26
	12.3 5. 練習問題	26
	12.4 6. 練習問題	26
	12.5 7. 練習問題	26
	12.6 8. 練習問題	26
	12.7 9. 練習問題	26
	12.8 10. 練習問題	26
<b>13</b>	付録	26
	13.1 perli / perl -de0	26
14	さいごに	27

## 1 はじめまして

## 1.1 はじめに

ここに来たということは、プログラミングを勉強しようと思って来たと思っています. プログラムで何かやろうと思っていることがあるかもしれませんし、まだそこまでイメージできていないけど、とりあえず勉強しよう. ということかもしれません.

プログラミングの世界へようこそ. サポータはあなたを手助けをします. このテキストは, 一人でも読めるように考えて書きますが, 必要であれば, 誰かに助けを求めてください. ここに居るならサポータが. 一人で読んでいるなら, インターネットの誰かが助けてくれるでしょう.

## 2 Perl について

## 2.1 Perl 言語

幸か不幸か、ここではプログラミング言語に Perl を選択しています. 少しだけ Perl 言語についてお話しします.

Perl 言語は,3分間で書き上げるようなやっつけ仕事にも,本格的なプログラムにも適しています.特に,テキスト処理が90%,その他が10% くらいの処理を書くのに適しています. 逆に,バイナリデータ $^1$  の処理は不得意です.

Perl は無料で手に入ります.

## 3 最初のプログラム

まずは、プログラムを書いてみて雰囲気を掴みましょう.

## 3.1 プログラムを書いて、実行してみる

- 作業用ディレクトリ (フォルダ) を作る
  - ディレクトリ名: perl20160220
- テキストエディタを開く
  - テキストエディタで作業フォルダ内にファイルを作る
  - ファイル名: 01\_hello.pl
- コンソール (コマンドプロンプト) を出して作業フォルダに移動する
  - 作業フォルダ内のファイルを実行する
  - 例: perl 01\_hello.pl
- 結果, コンソールに Hello, World と表示されたことを確認する

#### 3.1.1 実行例

- \$ mkdir perl20160220
- 2 \$ cd perl20160220
- 3 \$ edit 01\_hello.pl
- \$ perl 01\_hello.pl

 $<sup>^1</sup>$ バイナリデータとは、テキストではないデータです。つまり人が読めない形式のデータです。

#### 3.1.2 01\_hello.pl

```
#!/usr/bin/env perl

use strict;
use warnings;

print("Hello, World\n");
```

## 3.2 解説 - 周辺

まず、プログラムを実行するには、どのプログラムファイルを実行するかを 指定しなければなりません。どこにプログラムファイルを置いたかすぐわか るように、今日一日は、同じ作業ディレクトリで作業するようにしましょう。

perl 01\_hello.pl という呪文でプログラムを実行しました. これは, perl というコマンド(命令)を使って, 01\_hello.pl という Perl 言語で書かれた プログラムファイルを実行した. ということです. perl + プログラムのファイル名という形式は今後も使いますので, 慣れておいてください.

では、プログラムファイルの解説に移りましょう.

#### 3.3 解説 - プログラムファイル

Perl 言語は、上から順番に処理が進む、逐次処理を基本とします。プログラムファイルは、テキストファイルです。ソースコードやソース、ソースファイルとも呼ばれます。Perl 言語で書かれたソースコードの中では、#(シャープ)から行末まではコメント(説明文)の意味になります。コメントは、プログラムとしては何も実行されず、意味がありません。ですが、プログラムが何をしているかをソースコードを読む人に説明するために使います。

```
#!/usr/bin/env perl # <= Perl で動作するようにシェルに伝える

use strict; # 文法を厳密にチェックする
use warnings; # 警告を出す

print("Hello, World\n"); # 端末に文字列を出力する
```

最初の行は、Mac, Linux の人には意味があります。先程、perl + プログラムのファイル名 という形式で、Perl 言語のプログラムファイルを実行すると言いましたが、 実は、Mac や Linux の人には必要ありません. このプログラムファイルは、perl で実行する.というのはこの行に書かれています.

Windows の人は、ファイル名の拡張子部分で判定するようになっているので、この部分は意味がありません。

それ以外の環境の人は、その環境に詳しい人に聞いてください.

次の行の, use strict; use warnings; の部分は, 警告を多めに出すように指定してあります. 慣れないうち, 慣れていても, 情報が多い方が何か起きたときに気付きやすいので, この2行は基本的には付けるようにしましょう.

ここまでは、プログラムの動作にはあまり関係ありませんでした。残りの1行が、このプログラムファイルの本体です。

この行は、print() という関数を使って、コンソールに出力しています. プリントのくせに、プリンタへの出力ではありません. 出力した内容は、"Hello、World" であって、"Hello、World\n" ではありません. \n はどこへ行ってしまったのか. その部分を削除するなどして確かめてみてください.

; (セミコロン)が、1つの文の区切りに必要になります。

## 3.4 練習問題

- 一人でやってみよう
  - 他にも何かを表示してみよう(名前,年齢)
- perldoc -u -f atan2 を実行してみよう
- 以下のソースコードをファイルに保存して、実行して、結果を比較して みよう. 何が変わりましたか?

```
#!/usr/bin/env perl

dlines = `perldoc -u -f atan2`;

foreach (@lines) {
    s/\w<([^>]+)>/\U$1/g;
    print;
}
```

## 4 まずはプログラミングに慣れる

プログラミングといってもあまり怖がる必要はありません. 簡単なところから初めましょう.

#### 4.1 簡単な計算

プログラミング言語では、数値計算ができることが多いです. Perl でも、四 則演算など基本的な計算は可能です.

#### 4.1.1 02\_calc.pl

```
#!/usr/bin/env perl

use strict;
use warnings;

print(1 + 1, "\n"); # => 2

my $x = 10;
print($x * $x, "\n"); # => 10 * 10 => 100
```

## 4.2 解説 - 値 と変数

最初の行は、1+1 を計算した結果を、コンソールに出力しています。 print() は、コンソール出力をする命令でした。 1+1 は、ご存知の通り、数値計算です。 + という演算子は数値と数値を足して、新しい数値を作ります。この場合は、新しい数値は、print() 文によってコンソールに出力しています。

次の文 my sx = 10 は,sx という変数に数値 という値を代入しています.代入というのは値 に,名前を付けることです. この場合,名前はsx です. 代入の前に my が付いています.これは今はあまり気にしないでください.変数を最初に使う場所で使います.

10 に \$x\$ という名前を付けました. 次に  $print($x*$x, "\n")$  という文で,\$x\$ 掛ける \$x\$ を計算しています. \$x\$ は 10 でしたので,\$x\$ 10 の計算と同じ意味になります.

## 4.3 単語 - 値, 変数

#### 4.3.1 値, スカラ

新しい言葉がいくつか出てきたので、まとめておきます.

値とは、Perl で扱える最小単位のデータです。 スカラとも言います.

スカラ には、文字列と数値 があります. スカラの種類はそれ以外にもありますが、一番使うこれらを覚えておけばよいでしょう.

文字列 を作るには、以前出てきたように、ダブルクォートを使います。数値 を作るには、0 から 9 と . (ドット) などから作ることができます。

- スカラ.値
  - 文字列
    - \* 例. "\n", "Hello, World\n"
  - 数値
    - \* 例. 1, 2.5

#### 4.3.2 演算子

Perl では、数学の演算子をいくつか使うことができます. +, -, \*, / です. 掛け算、割り算の記号はなじみがないかもしれませんが、こういう記号を使います.

四則演算は,全て数値に対しての演算でした. 例えば,+ (足し算) は,2 つの数値を使って,1 つの数値を得るための演算です.

文字列 に作用する演算子もあります. . (ドット) です. これは 文字列 と 文字列 を結合して、新しい文字列 を作ります.

1 print("abc" . "def", "\n"); # => abcdef が表示される.

文字列 と 数値 の両方使う演算子もあります。 x は、文字列の繰り返しの意味の演算です。 文字列と繰り返しの数の数値 を使って、新しい文字列 を返します。

#### print("a" x 3, "\n"); # => aaa が表示される

- 演算子.
  - 数值,数值 →数值
    - \* 四則演算: +, -, \*, /
  - 文字列,文字列 → 文字列
    - \* 文字列連結: . (ドット)
  - 文字列,数值 → 文字列
    - \* 文字列の繰り返し: x

#### 4.3.3 関数呼び出し

print() について少し解説します。print() は、コンソールに出力する命令でした。このような命令は、関数と呼ばれます。関数は、関数名 + "(" + 引数 1 + "," + 引数 2… + ")" の形式で呼び出されます。今後もいくつか関数が出てきますが、基本は同じ呼び出し方です。(例外もあります)

print()は、すべての引数を連結してコンソールに出力します。

- 関数呼び出し
  - 例: print(1 + 1, "\n")

## 4.3.4 スカラ変数

変数は、値に名前を付けるための機能です. 最初に変数を使うときには、my を使います. 特にスカラに関連させる変数をスカラ変数と呼びます. スカラ変数は、"\$" + 変数名 の形式にします.

my には、変数の有効期間を設定する機能もありますが、今はあまり意識しなくともよいでしょう.

- 変数
  - my で宣言する
  - スカラ変数
    - \* \$ + 変数名 の形式で使う
    - \* 変数 = 値で、変数に値をひもづけます.
      - · 例: \$num = 10, \$greeting = "おはよう"
    - \* 変数 の形式で変数にひもづいた値を使えます
      - · 例: \$x = 10; \$x + 1; # => 11

## 4.4 練習問題

- いくつか計算してみましょう. 例 1 + 2 \* 3. 答えは思い通りになりましたか?
- print 文を複数使って、プロフィールを表示してみてください. (名前, 誕生日, ...)
- 半径 12.5 の円の円周の長さを求めるプログラムを書いてください. 円 周の長さは半径の  $2\pi$  倍 です. 答えは, 78.5 くらいになる筈です.

## 5 今回目指すもの

この会では、TODO リストを作りながら、一緒にプログラミングについて学びます.まず、何を作るかです.

#### 5.1 今回目指すもの

TODO リストプログラムを作ります. どんな機能があるでしょうか? 私達の作る TODO リストは、以下のような機能を持っています.

- TODO リストとは、TODO を保存、操作します
- TODO の状態は、未実施、完了 があります.
- TODO には内容 (content) があります.

#### 5.2 練習問題

• 自分でも TODO リストに欲しい機能を考えてみましょう

## 6 TODO リストを保存する

ではまず、固定の TODO リストを作成するプログラムを作ってみましょう.

## $6.1 \quad 03_{init\_todo.pl}$

```
#!/usr/bin/env perl

use strict;
use warnings;

open(my $fh, ">", "todolist.txt") or die;
print($fh "sample todo,Done\n");
close($fh) or die;
```

新しい関数が出てきました. 順に説明します.

## 6.2 open() / close() 関数

ファイルの中身を読み込んだり、書き込んだりするための準備や後片付けが必要です。それを行うのが、open()、close() 関数です。

open() 関数は, 引数 が3つあります.

- 1つめは、ファイルハンドル、my \$fh の部分です. ファイルハンドル というのは、ファイルを扱うためのスカラ値です. print() 関数のと ころでもう少し説明します.
- 2つめは、モードです、書き込みか読み込みかを文字列で指定します、
  - ">" は書き込み
  - "<" は読み込みで、新規ファイルを作って先頭から書き込みます.
  - ">>" は、追記書き込みで、ファイルの末尾に追記します.
- 3つめは,ファイル名です.
  - ファイル名だけ書くと、プログラムを実行したディレクトリと同 じ場所にあるものになります.

close() 関数は、ファイルハンドルの後片付けをします。忘れないようにしましょう。

ファイルの書き込み読み込みは、以下のような形が基本です. 覚えてしまいましょう.

書き込み

```
open(my $fh, ">", "filename") or die;

# 何か処理をする

close($fh) or die;
```

読み込み

```
open(my $fh, "<", "filename") or die;

# 何か処理をする

close($fh) or die;
```

or die については、ちょっと難しいので、とりあえずこういうものとして覚えておいてください.

## 6.3 print() 関数

print()です。これはコンソールに出力する関数だった筈です。実は、それ以外の機能もあります。ファイルハンドルを指定すると、そのファイルに出力してくれます。このときに、ファイルハンドルを指定した後には、カンマがないことに注意してください。

```
print("Hello, World\n");
print($fh "Hello, World\n"); # カンマが無いことに注意
```

## 6.4 練習問題

- todolist.txt ファイルの内容を確認してみましょう.
- ファイルに色々書き込んでみましょう.

## 7 TODO を追加する

TODO を追加する機能を追加しましょう.

## 7.1 04\_add\_todo.pl

```
#!/usr/bin/env perl
use strict;
use warnings;
```

```
my $content = <STDIN>;
chomp($content);
print($content, "\n");

open(my $fh, ">>", "todolist.txt") or die;
print($fh $content . ",NotYet", "\n");
close($fh) or die;
```

#### 7.2 行入力演算子

<> (この場合だと<STDIN>) は、演算子です. 今までに出てきたものとちょっと使い方が違います.

今までの演算子の使い方の例.

```
1 1 + 2; # => 3 : 数值,数值 → 数值
2 "a" x 3; # => aaa 文字列,数值 → 文字列
```

今までに出てきた演算子は、全て演算子の両側に引数を取っていました. <> は、< + ファイルハンドル + > のように真ん中に引数を書きます.

ですので、<STDIN>の引数は、STDIN です。 STDIN はファイルハンドルです。open() 関数のところで出てきたものと同じです。 STDIN は、コンソールからの入力が溜められているファイルハンドルです。

```
my $line;

sline = <STDIN>;
fline = <$fh>;
```

- ファイルハンドル → 文字列
  - 行入力演算子: <>

#### 7.3 スカラの自動変換

Perl は、スカラ値の自動変換が行われます.

```
1 # 文字列 → 数値 への自動変換 ("10" → 10)
2 1 + "10"; # => 2
3
```

```
4 #数値 文字列の自動変換 (3 → "3")
```

5 3 x 4; # => 3333

Perl がうまくやってくれるので、こういうことが起きている. ということだけ知っておいてください.

## 7.4 chomp() 関数

chomp() は、末尾の改行を取り除きます.変数を指定すると、その変数にひもづいている値が変更されます.

## 7.5 練習問題

- プログラムを実行してみてファイルの中身を見て TODO が追記されて いることを確認しましょう
- 何個かの入力をして、計算をするプログラムを書いてみてください. 例 えば、2つの数を入力してもらって、足し算をするプログラムなどです
- 半径を入力したら、直径を表示するプログラムを作ってください

## 8 現在 の TODO を表示する

TODO を追加するところまでできるようになりました. では、現在の TODO 全部を表示できるようなプログラムを作りましょう.

## $8.1 \quad 05\_list\_todo.pl$

```
#!/usr/bin/env perl

use strict;
use warnings;

open(my $fh, "<", "todolist.txt") or die;
my $count = 1;
while (defined(my $line = <$fh>)) {
```

```
9     print($count, ":", $line);
10     $count = $count + 1;
11   }
12   close($fh) or die;
```

## 8.2 defind() とファイルの終端

<> 行入力演算子は、ファイルの終端まで入力が進むと、undef という値を返す. これもスカラ値です.

defined() 関数は、引数が undef かどうか判定してくれて、真偽値を返してくれます. 真偽値については、次の while の条件部分のところで触れます.

#### 8.3 while 文

while というのは、とある条件が満たされている間に、何か別のことを行う. というときに使います.

例えば、while (A) { B } のようになっていたら、最初に A が満たされているかチェックし、満たされていれば、B を処理し、次に A をチェック、B を処理… で A が満たされなくなったときに、この繰り返しが終了し、次の行に処理が移動します.

条件が満たされている状態を真,そうでない状態を偽と呼びます. Perl では,偽とはスカラ値で言うと,undef,数値の 0,文字列の "" (空文字列) です. 又,文字列"0" は,数値 0 に自動変換されるので,これも偽です. ""空文字列に自動変換される数値はありませんので,逆はありません. 真は,それ以外で表現できます.

- ブーリアン (スカラ版)
  - 偽: undef, 0, "", "0"
  - 真: それ以外

例えば,

```
while (1) {
print("hello", "\n"); # 実行する前に説明を読んでね
}
```

この場合は、条件部分が 1 となっていて、これは数値の 0 ではないので、常に真です。つまり print() 文を処理し、条件に戻っても、常に真なので、print() が無限に繰り返されることになります。

無限に繰り返すのはあまり良くないので、回数を決めたりすることがよくあります.

例えば,

```
my $count = 1;
while ($count < 10) {
    print($count, "\n");
    $count = $count + 1;
}</pre>
```

こうすると、最初は \$count in 0 ですが、ブロックの中の\$count = \$count + 1 が実行されるたびに \$count の値が増えていくので、<math>\$count in 10 になったら、繰り返しが終了します.

## 8.4 ブーリアン演算子

ブーリアンを扱う演算子がいくつかあります. 例えば数値に対してだと, <, > などの記号は見たことがあるのではないでしょうか?これらは, ご存知の通りに数値の大小でブーリアンとして適切なスカラ値を返します.

- 演算子
  - 数値,数値 → ブーリアン<sup>2</sup>
     \* <, >, <=, >=, ==, !=
     文字列,文字列 → ブーリアン
     \* eq, ne
     ブーリアン,ブーリアン → ブーリアン
     \* and, or, &&, ||
     ブーリアン → ブーリアン
     \* not,!

## 8.5 練習問題

• 繰り返しを実行してみましょう.

## 9 TODO の内容のうち未完了のものだけ表示する

終了した TODO は見る必要が無いと思います. では, TODO のうち未完了のものだけ表示するプログラムを作りましょう.

## 9.1 06\_list\_notyet\_todo.pl

```
#!/usr/bin/env perl

use strict;
use warnings;

open(my $fh, "<", "todolist.txt") or die;
my $count = 1;
while (my $line = <$fh>) {
    chomp($line);
    my ($content, $state) = split(/,/, $line);
    if ($state eq "NotYet") {
        print($count, ":", $line, "\n");
    }

$count = $count + 1;
}
close($fh) or die;
```

## 9.2 リスト / リスト変数

単一のデータとしてスカラ値があるという話をしました。複数のスカラ値をまとめて取り扱うために、リストという値があります。

```
1 (1, 2, 3);
```

リストは、() (括弧) で囲まれて、、(カンマ) で区切られたスカラ値で表 現します。

スカラにスカラ変数があったように、リストにもリスト変数があります。リスト変数は、**c** + 変数名です。スカラ変数と同じように、myで宣言して、=で代入できます。

```
my @numbers = (1, 2, 3, 4, 5);
```

リストには、複数のスカラ値が含まれています。個別のスカラ値を取り扱うために、[]を使います。

```
my @numbers = (1, 2, 3, 4, 5);
snumber[0]; # => 1
snumber[2]; # => 3
```

リスト変数全体を表すときには、先頭の記号が、0 ですが、リストの中のスカラ値部分を扱う場合には、 先頭の記号が\$ になります. [] 内の数値 (添字) は、リストの先頭からの位置で場所を指定します. 先頭の位置は、0 です. リストの一部を代入して変更することもできます.

```
ny @numbers = (1, 2, 3, 4, 5);

2 $number[0] = 10; # => @numbers は (10, 2, 3, 4, 5)
```

- 変数
  - リスト変数
    - \* @ + 変数名
    - \* \$ + 変数名 + [ + 位置 + ]

## 9.3 split() 関数

split()は、区切り文字と文字列を引数に取り、文字列を分割します。 文字列を分割して、文字列のリストを返します。

## 9.4 スカラコンテキスト / リストコンテキスト

コンテキストは、その位置に置かれたものは、何として扱われるか、という ことです、まったくわからないかと思いますので、例を上げます.

```
# スカラーコンテキスト
my $line = <>;

# リストコンテキスト
my ($line1, $line2, $line3) = <>;
```

<> 行入力演算子に何を求めているか,でコンテキストが違う.と表現します.例えば,上の例で my \$line = <> の方は,代入の相手がスカラ変数な

ので、<> もスカラを返そうとします.これがスカラコンテキストです.又、my (\$1ine1, \$1ine2, \$1ine3) = <> の方は、代入の相手はスカラではありません。<> に複数の値を求めています.リストコンテキストと呼ばれます.

見分け方としては、代入先が()(括弧)を使っているか、リスト変数か否かです。

```
my ($line) = <>;
```

これは,最初の例と同じように見えますが,()を使っているのでリストコンテキストになります.

上記の例は、代入でしたが、while などの条件式や演算子の引数もスカラコンテキストです。

- コンテキスト
  - スカラコンテキスト
    - \* スカラ変数への代入,条件式,演算子の引数部分
  - リストコンテキスト
    - \* それ以外

## 9.5 練習問題

- 半径を入力して直径を出力を作りましたが、半径が負の数だったら、直径として 0 を表示するようにしてください
- 以下のプログラムがどうなるか予想して、実際に確認してみてください

```
if ("0") {
    print("I heard truth", "\n");
}
```

## 10 TODO を完了済みにする

TODO を完了させられなければ、TODO リストとしては未完成です。 TODO を指定して、その TODO を完了済み (Done) としてマークできるようにしましょう.

以前に作った、未完了の TODO を表示するプログラムからの出力に出なくなる筈です.

## 10.1 07\_done\_todo.pl

```
#!/usr/bin/env perl
   use strict;
   use warnings;
   print("which number?: ");
   my $num = <STDIN>;
   open(my $rfh, "<", "todolist.txt") or die;</pre>
   my @lines = <$rfh>;
   close($rfh) or die;
   open(my $wfh, ">", "todolist.txt") or die;
   my $count = 1;
   foreach my $line (@lines) {
       chomp($line);
16
       my ($content, $state) = split(/,/, $line);
       if ($num == $count) {
18
            $state = "Done";
20
       print($wfh $content, ",", $state, "\n");
21
       $count = $count + 1;
22
   close($wfh) or die;
```

## 10.2 if 文

while と同じように、条件とブロックがあります. while と違い繰り返したりしません. 条件式が真であればブロックを実行, そうでなければ, ブロックを実行しません.

```
my $input = <STDIN>;
if ($input) {
    print("Your input(", $input, ") is true", "\n");
4 }
```

## 10.3 練習問題

• 1行に1つずつの数のリストを入力の終りまで読み込んで、以下に示す 人名のリストから数に対応する人名を表示するプログラムを書いてく ださい. 例えば、入力された番号が1,2,4,2 だとすると、fread、betty、 dino, betty となります.

fred betty barney dino wilma pebbles bamm-bamm

## 11 まとめ

TODO リストを操作するプログラムを作ってみました. そのプログラムの解説をすることで、Perl プログラムの基礎を学びました.

この会で Perl プログラミングについて学んだことをまとめます

#### 11.1 準備

- 作業ディレクトリを作って、そこで作業した
  - ファイル名を指定すると、作業ディレクトリのファイルを指定したことになる
- テキストエディタで Perl のプログラムファイルを作ることができる
- Perl プログラムファイルをコンソールから実行できる

## 11.2 基礎文法

- ファイル名は, .pl で終わるようにする
- #(シャープ)から行末までは、コメントとして扱われてプログラムと しては実行されない
- 行の区切りに; (セミコロン) を付ける

• 下記の基本の行を必ず書く

```
#!/usr/bin/env perl
use strict;
use warnings;
```

## 11.3 スカラ値

- 1つ分のデータのことをスカラと呼ぶ
- スカラにはいくつかの種類がある
  - 数値
  - 文字列
  - ファイルハンドル
  - undef

```
1 # 数値の例
  1
  10
  2.5
  -100
  # 文字列の例
  "文字列"
  "\n"
  "Hello, World"
10
  # ファイルハンドルの例
  open(my $fh, "<", "filename") or die; # この内の $fh に入っている
13
  STDIN
15
  # undef の例
16
  my $var; #変数を宣言だけして、代入する前には undef になっている
  $var = undef;
```

又,場所によってスカラ値はブーリアン,真偽値としても扱われる.

- ブーリアン, 真偽値
  - 偽: 0, "", undef, "0"
  - 真: それ以外

```
1 # 偽
2 if (0) {
     print("I said truth", "\n");
4 }
  if ("") {
      print("I said truth", "\n");
  }
   if (undef) {
      print("I said truth", "\n");
  }
10
 # 真
13 if (1) {
  print("I heard truth", "\n");
  }
16 if ("a") {
  print("I heard truth", "\n");
```

## 11.4 リスト値

複数のスカラをまとめて扱う.

## 11.5 変数

## 11.5.1 スカラ変数

- スカラ変数はスカラに名前を付ける
- 最初に使うときには my をつける
- スカラ変数にスカラをひもづけるときには、= (イコール)を使う.
  - 代入と言う

## 11.5.2 リスト変数

- リスト値に名前を付ける
- 最初に使うときには, my をつける

- 代入には, = を使う
- それぞれのスカラ値を取り扱うときには, [] を使う
  - 例: \$ary[0], \$ary[0] = 10

## 11.6 演算子

引数を取って,新しい値を返す.

- 数值,数值 →数值
  - 四則演算: +, -, \*, /
- 文字列,文字列 → 数值
  - 文字列連結: .
- 文字列,数值 → 数值
  - 文字列繰り返し: x
- ファイルハンドル → 文字列
  - 行入力演算子: <>
- 数値,数値 → ブーリアン
  - <, >, <=, >=, ==, !=
- 文字列, 文字列 → ブーリアン
  - eq, ne
- ブーリアン, ブーリアン → ブーリアン
  - and, or, &&, | |
- ブーリアン → ブーリアン
  - not, !

## 11.7 関数

関数は, 関数名 + "(" + 引数1 + ", " + 引数2 ... ")" の形式で呼び出す. print()

引数をコンソールに出力する最初の引数をファイルハンドルにすると, そのファイルハンドルに出力する

open()

指定したファイルのファイルハンドルを作る "<" 読み込み, ">" 書き込み, ">>" 追記のモードがある.

close()

ファイルハンドルの後片付けをする

chomp()

引数の値の末尾の改行文字を削除する

split()

文字列を区切りで区切って, リスト値にして返す

## 11.8 while 文

while では繰り返し処理を行うことができます. 条件式の中は, ブーリアンとして解釈され, 条件式が真の間, ブロックの中を実行し続けます.

```
my $count = 0;
while ($count < 5) {
    print($count, "\n");
    $count = $count + 1;
}</pre>
```

## 11.9 if 文

if は、ブロックを実行するかしないかを、条件式で決めます.

```
my $a = 10;
if ($a > 5) {
    print("over", "\n");
}
```

#### 11.10 コンテキスト

コンテキストによって,動作を変える演算子,関数がある.

- コンテキスト
  - スカラコンテキスト \* スカラへの代入,条件式部分,演算子の引数
  - リストコンテキスト
    - \* それ以外

例えば,以下があります.

- <> 行入力演算子
  - スカラ: 1 行分だけ返す
  - リスト: 全部の行をリストにして返す

## 12 練習問題の答え

- 12.1 3. 練習問題
- 12.2 4. 練習問題
- 12.3 5. 練習問題
- 12.4 6. 練習問題
- 12.5 7. 練習問題
- 12.6 8. 練習問題
- 12.7 9. 練習問題
- 12.8 10. 練習問題
- 13 付録
- 13.1 perli / perl -de0

簡単に調査したいときに毎回ファイルを作って実行するのは面倒ですよね. 例えば、perli というコマンドがあります.

perli のダウンロード

- \$ curl -LO https://raw.githubusercontent.com/mklement0/perli/master/bin/perli
- \$ chmod +x perli
- 3 \$ ./perli # 又は, perl perli

perli は、Perl プログラムなので、ダウンロードしたらそのまま使うことができます. perli を実行すると、簡単に値を調べることができます.

```
1  > 1
2  1
3  > defined(undef)
4  ''
5  > defined(10)
6  1
7  > not 1
8  ''
9  > (1, 2, 3)
10  1
11  2
12  3
```

色々試してみてもいいかもしれません.

同じ用途で、perl -de0 というコマンドも使えますが、若干使い方が難しいです. こちらは、perl があれば最初から使えるでしょう.

```
1  $ perl -de0
2  DB<1> x 1
3  0  1
4  DB<2> x defined(undef)
5  0  ''
6  DB<3> x (1,2,3)
7  0  1
8  1  2
9  2  3
```

x を先頭に付けてコマンドを実行します. 結果はリスト値が添字と一緒に順 に縦に並びます.

## 14 さいごに

練習問題は、『初めての Perl』 のものをお借りしています.素晴しいテキストありがとうございます.これを読んだ人がプログラミングを挫折しないで学べるようになることを望んで書きました.